Pensieve header: Implementing Θ - the main notebook accompanying Talks/UBC-241004..

exec

```
nb2tex$TeXFileName = "Theta.tex";
```

pdf

Preliminaries

pdf

This is Theta.nb of http://drorbn.net/ubc24/ap.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\UBC-241004"];
```

pdf

In[∘]:=
```
Once[<< KnotTheory`; << Rot.m; << PolyPlot.m];
```

pdf

```
C:\drorbn\AcademicPensieve\Projects\KnotTheory\KnotTheory
```

pdf

```
Loading KnotTheory` version of September 27, 2024, 13:23:33.5336.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
Loading Rot.m from http://drorbn.net/ubc24/ap to compute rotation numbers.
```

pdf

```
Loading PolyPlot.m from http://drorbn.net/ubc24/ap to plot 2-variable polynomials.
```

pdf

The Program

pdf

In[∘]:=
```
CF[ℰ_] := Module[{vs = Union@Cases[ℰ, g__, ∞], ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :> Factor[c] (Times @@ vs^ps)]];
```

pdf

In[∘]:=

$$T_3 = T_1\, T_2;$$

pdf

In[∘]:=

$$\begin{aligned}&R_1[s\_,\, i\_,\, j\_] =\\ &\quad \mathrm{CF}\Big[s\,\Big(1/2 - g_{3ii} + T_2^s\, g_{1ii}\, g_{2ji} - g_{1ii}\, g_{2jj} - \left(T_2^s - 1\right) g_{2ji}\, g_{3ii} + 2\, g_{2jj}\, g_{3ii} - \left(1 - T_3^s\right) g_{2ji}\, g_{3ji} -\\ &\qquad g_{2ii}\, g_{3jj} - T_2^s\, g_{2ji}\, g_{3jj} + g_{1ii}\, g_{3jj} + \Big(\left(T_1^s - 1\right) g_{1ji}\, \left(T_2^{2\,s}\, g_{2ji} - T_2^s\, g_{2jj} + T_2^s\, g_{3jj}\right) +\\ &\qquad\quad \left(T_3^s - 1\right) g_{3ji}\, \left(1 - T_2^s\, g_{1ii} - \left(T_1^s - 1\right) \left(T_2^s + 1\right) g_{1ji} + \left(T_2^s - 2\right) g_{2jj} + g_{2ij}\right)\Big) \Big/ \left(T_2^s - 1\right)\Big)\Big];\end{aligned}$$

pdf

In[∘]:=

$$\begin{aligned}&\Theta[\{s0\_,\, i0\_,\, j0\_\},\, \{s1\_,\, i1\_,\, j1\_\}] := \mathrm{CF}\Big[s1\, \left(T_1^{s0} - 1\right) \left(T_2^{s1} - 1\right)^{-1}\\ &\quad \left(T_3^{s1} - 1\right) g_{1,j1,i0}\, g_{3,j0,i1}\, \left(\left(T_2^{s0}\, g_{2,i1,i0} - g_{2,i1,j0}\right) - \left(T_2^{s0}\, g_{2,j1,i0} - g_{2,j1,j0}\right)\right)\Big]\end{aligned}$$

pdf

In[∘]:=

$$\Gamma_1[\varphi\_,\, k\_] = -\varphi/2 + \varphi\, g_{3kk};$$

pdf

In[∘]:=
```
Θ[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ν, α, β, gEval, c, z},
  {Cs, φ} = Rot[K]; n = Length[Cs];
  A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1
                                                              0     -1    ))];
  Δ = T^((-Total[φ]-Total[Cs⟦All,1⟧])/2) Det[A];
  G = Inverse[A];
  gEval[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :→ (G⟦α, β⟧ /. T → T_ν)];
  z = gEval[∑_{k1=1}^n ∑_{k2=1}^n θ[Cs⟦k1⟧, Cs⟦k2⟧]];
  z += gEval[∑_{k=1}^n R_1 @@ Cs⟦k⟧];
  z += gEval[∑_{k=1}^{2 n} Γ_1[φ⟦k⟧, k]];
  {Δ, (Δ /. T → T_1) (Δ /. T → T_2) (Δ /. T → T_3) z} // Factor ];
```

The Rolfsen Table

In[∘]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

KnotTheory: Loading precomputed data in PD4Knots`.

In[∘]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 25], Spacings → 0, Dividers → All]
```

Out[∘]=



In[•]:=
```
Export["g250.png", g250]
```
Out[•]=
```
g250.png
```

pdf

The Trefoil, Conway, and Kinoshita-Terasaka

pdf

In[•]:=
```
Θ[Knot[3, 1]] // Expand
```
Out[•]=
pdf

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2\,T_2^2}+\frac{1}{T_1\,T_2^2}+\frac{1}{T_1^2\,T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2\,T_2-T_2^2+T_1\,T_2^2-T_1^2\,T_2^2\right\}$$

pdf

In[•]:=
```
GraphicsRow[PolyPlot[Θ[Knot[#]]] & /@ {"3_1", "K11n34", "K11n42"}]
```
pdf

KnotTheory: Loading precomputed data in DTCode4KnotsTo11`.

pdf

KnotTheory: The GaussCode to PD conversion was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.

Out[•]=
pdf



tex

(Note that the genus of the Conway knot appears to be bigger than the genus of Kinoshita-Terasaka)

pdf

Some Torus Knots

pdf

In[∘]:=
```
GraphicsRow[PolyPlot[Θ[TorusKnot @@ #]] &
  /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}, Spacings → Scaled@0.05]
```

Out[∘]=
pdf



The 132-crossing Torus Knot $T_{22/7}$

In[∘]:=
```
AbsoluteTiming[T227 = Θ[TorusKnot[22, 7]];]
PolyPlot[T227]
```

Out[∘]=

{702.377, Null}

Out[∘]=



In[ ]:=
```
Export["T227.pdf", PolyPlot[T227]]
```
Out[ ]=
```
T227.pdf
```

Large Random Knots

See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[ ]:=
```
DK300 = PolyPlot@Expand[
   Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D300.m"][[
     2]] /. {T1 → T1, T2 → T2}]
```
Out[ ]=



In[ ]:=
```
Export["DK300.png", DK300, ImagePadding → None, PlotRangePadding → None]
```
Out[ ]=
```
DK300.png
```

In[ ]:=
```
Export["DK300.pdf", DK300]
```
Out[ ]=
```
DK300.pdf
```

Invariance under R3

exec
```
nb2tex$TeXFileName = "Invariance.tex";
```

pdf

In[•]:=
```
δ_{i_,j_} := If[i === j, 1, 0];
gR_{s_,i_,j_} := {g_{ν_ iβ_} :> δ_{iβ} + T_ν^s g_{ν i⁺ β} + (1 - T_ν^s) g_{ν j⁺ β},
   g_{ν_ jβ_} :> δ_{jβ} + g_{ν j⁺ β}, g_{ν_ α_ i} :> T_ν^{-s} (g_{ν α i⁺} - δ_{α i⁺}), g_{ν_ α_ j} :> g_{ν α j⁺} - (1 - T_ν^s) g_{ν α i} - δ_{α j⁺}}
```

Proof of Reidemeister 3:

pdf

In[•]:=
```
DSum[Cs___] := Sum[R_1 @@ c, {c, {Cs}}] + Sum[θ[c0, c1], {c0, {Cs}}, {c1, {Cs}}]
lhs = DSum[{1, j, k}, {1, i, k⁺}, {1, i⁺, j⁺}, {s, m, n}] //. gR_{1,j,k} ⋃ gR_{1,i,k⁺} ⋃ gR_{1,i⁺,j⁺};
rhs = DSum[{1, i, j}, {1, i⁺, k}, {1, j⁺, k⁺}, {s, m, n}] //. gR_{1,i,j} ⋃ gR_{1,i⁺,k} ⋃ gR_{1,j⁺,k⁺};
Simplify[lhs == rhs]
```

Out[•]=
pdf

True

Some Virtual Knots

In[ ]:=
```
pd = PD[X[2, 4, 3, 1], X[3, 1, 4, 2]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[ ]=

{{{-1, 4, 2}, {-1, 1, 3}}, {0, 0, 1, 0}}

Out[ ]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{-1-2\,T_2+4\,T_1\,T_2}{2\,T_1\,T_2}\right\}$$

Out[ ]=



In[∘]:=
```
pd = PD[X[3, 8, 4, 1], X[1, 6, 2, 7], X[5, 2, 6, 3], X[7, 5, 8, 4]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[∘]=

{{{-1, 8, 3}, {-1, 6, 1}, {-1, 2, 5}, {1, 4, 7}}, {0, 0, 0, 0, 1, -1, -1, -1}}

Out[∘]=

$$\left\{-\frac{1-3T+T^2}{T},\ -\frac{\left(1-3T_1+T_1^2\right)\left(-1+T_1+2T_1^2+T_2-7T_1^2T_2-2T_1^3T_2-T_1T_2^2+4T_1^2T_2^2+5T_1^3T_2^2-3T_1^2T_2^3+T_1^3T_2^3\right)}{T_1^2}\right\}$$

Out[∘]=



In[ ]:=
```
pd = PD[X[2, 8, 3, 1], X[3, 1, 4, 2], X[6, 4, 7, 5], X[7, 5, 8, 6]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[ ]=

{{{-1, 8, 2}, {-1, 1, 3}, {-1, 4, 6}, {-1, 5, 7}}, {0, 0, 1, 0, 0, 1, 1, 0}}

Out[ ]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{1}{2 T_1^3 T_2^4}\left(4 - 4 T_1 - 6 T_2 + 2 T_1 T_2 + 6 T_1^2 T_2 + 4 T_1 T_2^2 - 8 T_1^2 T_2^2 + 2 T_2^3 + 2 T_1 T_2^3 + T_1^2 T_2^3 - 4 T_1^3 T_2^3 - 2 T_1 T_2^4 - 6 T_1^2 T_2^4 + 10 T_1^3 T_2^4\right)\right\}$$

Out[ ]=